
Appendix B. Subroutine WGEOM for the Hemispherical Helix

The following parameters in WGEOM subroutine are used for a hemispherical

helix.

NSEGSP Total number of wire segments on the hemisphere

ZOFFSET Distance the helix is offset from the ground plane in meters

NTOTAL Total number of turns that sphere would have if wound fully

NACT Actual number of turns on the hemispherical helix

AS Radius of hemisphere in meters

These data are included at the end of the input file for the READ statements in the

ESP.  Simple data for a 4.5-turn hemispherical helix with a radius of 0.02 meters and

with an offset distance of 0.005 meters from the ground plane are presented below.

The WGEOM subprogram for the hemispherical helix is listed below.

SUBROUTINE WGEOM(IA,IB,X,Y,Z,NM,NP,NAT,NSA,NPLA,VGA,BDSK,ZLDA,
     +NWG,VG,ZLD,WV,NFS1,NFS2,NRUN,A)
      DIMENSION IA(1),IB(1),X(1),Y(1),Z(1),
     +NSA(1),NPLA(1),BDSK(1)
      COMPLEX VGA(1),ZLDA(1),VG(1),ZLD(1)
      REAL PIE, AS, NTOTAL, NACT
C
C  Total number of wire segments on sphere, NSEGSP
      READ(5,*)NSEGSP

C
C  Length of segment offsetting from ground plane, ZOFFSET
C  If ZOFFSET is not zero there will be one additional segment
      READ(5,*)ZOFFSET
C
      IF (ZOFFSET .GT. 0.0) INDX = 1
      NM = NSEGSP + INDX
C  Total number of wire points, NP
      NP = NM + 1
C  Total number of wire attachment points, NAT
      NAT = 1
C
C  The Spherical Helix geometry is now defined
C
C  Total number of turns sphere would have if completely wound, NTOTAL
      READ(5,*)NTOTAL
C
C  Actual number of turns sphere has, NACT
      READ(5,*)NACT
C
C  Radius of sphere, AS (meters)
      READ(5,*)AS
C
      PIE = 3.1415927
C  SEGPHI is the change in the angle phi (radians) for each segment
      SEGPHI = 2.0*PIE*NACT/REAL(NSEGSP)
C
C  Define coordinates of wire points and endpoints of segments
      X(1) = 0.0
      Y(1) = 0.0
      Z(1) = 0.0
      IA(1) = 1
      IB(1) = 2
        IF (INDX .EQ. 0) GOTO 101
      X(2) = 0.0
      Y(2) = 0.0
      Z(2) = ZOFFSET
      IA(2) = 2
      IB(2) = 3
  101 DO 100 J = 2+INDX,NP
        PHI = SEGPHI*REAL(J-1-INDX)
        TH = ACOS((PHI/(PIE*NTOTAL))-0.00000005)
        X(J) = AS*SIN(TH)*COS(PHI)
        Y(J) = AS*SIN(TH)*SIN(PHI)
        Z(J) = AS*COS(TH)+ ZOFFSET
          IF (J .EQ. NP) GOTO 100
        IA(J) = J
        IB(J) = J+1
  100 CONTINUE
C
C  Check that smallest segment is greater than twice the wire radius
      SLENGTH = SQRT( (X(NP)-X(NP-1))**2 + (Y(NP)-Y(NP-1))**2 +
     +          (Z(NP)-Z(NP-1))**2 )
      IF (SLENGTH .LE. A*2.) WRITE(*,*)'   WARNING - At least one segmen
     +t is smaller than twice the wire radius'
C

C  Wire "location" of attachment point, NSA
      NSA(1) = 1
C  Plate number geometry is attached to, NPLA
      NPLA(1) = 1
C  Complex voltage generator, VGA
      VGA(1) = (1.0,0.0)
C  Outer disk radius of disk monopole (should be about .2lambda), BDSK
      BDSK(1) = 0.2*WV
C  No mutual coupling computations
C     NFS1=0
C     NFS2=0
C
      RETURN
      END

